# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Simon Hanna <simon dot hanna AT serve-me DOT info>

pkgname=python-zope-event
pkgver=5.0
pkgrel=1
pkgdesc="Provides a simple event system"
arch=(any)
license=('ZPL')
url="https://github.com/zopefoundation/zope.event"
depends=('python')
makedepends=('python-setuptools')
checkdepends=('python-zope-testrunner')
source=("https://github.com/zopefoundation/zope.event/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha512sums=('df1719f87e68fafe83b775293bd06a26ab82c1d75e7106fe81f04dc8ec8fc43d966e05171b50052e7c1bdf675b0882fba8d71f087f4756f89ded91a3fb94782c')

build() {
  cd zope.event-$pkgver
  python setup.py build
}

check() {
  cd zope.event-$pkgver
  PYTHONPATH="$PWD/src" python -m zope.testrunner --test-path=src
}

package() {
  cd zope.event-$pkgver
  python setup.py install --root="$pkgdir/" --optimize=1
}

# vim:set ts=2 sw=2 et:
